

Working with IF Functions

The **IF** function is one of a collection of *logical* functions whose actions depend on a number of conditions that you set. An **IF** function requires three arguments. The syntax is:

IF(logical_test, [value_if_true], [value_if_false])

The *logical_test* is a question or circumstance than can be answered *yes* or *no*, or *true* or *false*.

The *value_if_true* is an expression describing what the function should do if the answer to the test is *yes* or *true*.

The *value_if_false* is an expression describing what the function should do if the answer to the test is *no* or *false*.

In other words, you ask a question. If the answer is yes you do one thing. If the answer is no you do something else. What the IF statement does could be to display a number or piece of text, or it can do a calculation. If you do not supply a *value_if_true* or *value_if_false* argument (these arguments are optional as indicated by the square brackets) the function will return *TRUE* or *FALSE* depending upon whether or not the logical test has been satisfied.

A Simple IF Function

This example supposes that the tonnages for wheat have been overestimated and need to be reduced by 5%. Tonnages for other commodities can remain unchanged. The *logical_test* is that the value in the *Commodity* column (cell E2) is *Wheat*. If that is *true* then the function returns the value in the corresponding row in the *Tonnes* column (cell F2) multiplied by 95%. If the test returns false (i.e. the value in E2 is not *Wheat*) then the function simply returns the value from cell F2 without any changes. Note that the *Tonnes* values in rows 2 and 4 in the screenshot have been changed (Fig.1).

		G2					fx =IF(E2="Wheat",F2*95%,F2)	
	A	B	C	D	E	F	G	
1	Year	Month	CountryFrom	CountryTo	Commodity	Tonnes	Adjusted Value	
2	2000	1	USA	United Kingdom	Wheat	16182	15372.9	
3	2005	2	USA	United Kingdom	Rice	11398	11398	
4	2002	6	Mexico	Portugal	Wheat	17497	16622.15	
5	2005	3	Canada	Netherlands	Soybeans	15627	15627	
6	2004	9	Brazil	Italy	Maize	10748	10748	

Fig.1 A simple IF Function.

Adding a Condition to the Test

If an additional condition has to be added to the test this can be done using the **AND** function. This example supposes that the tonnages for wheat have been overestimated but only during the year 2002. The *logical_test* argument uses the **AND** function to supply both the conditions that the value in the *Commodity* column (cell E2) must be *Wheat* and that in the *Year* column (cell A2) must be *2002*. Note that the *Tonnes* value row 2 in the screenshot is unchanged because the year is not 2002, but that in row 4 belongs to 2002 and has been changed (Fig.2).

		G2					fx =IF(AND(E2="Wheat",A2=2002),F2*95%,F2)	
	A	B	C	D	E	F	G	
1	Year	Month	CountryFrom	CountryTo	Commodity	Tonnes	Adjusted Value	
2	2000	1	USA	United Kingdom	Wheat	16182	16182	
3	2005	2	USA	United Kingdom	Rice	11398	11398	
4	2002	6	Mexico	Portugal	Wheat	17497	16622.15	
5	2005	3	Canada	Netherlands	Soybeans	15627	15627	
6	2004	9	Brazil	Italy	Maize	10748	10748	

Fig.2 The logical test contains two conditions.

Adding an Alternative Condition

In the previous example both of the conditions of the test had to be satisfied. This example adds a condition to the test that is an alternative. It uses the **OR** function to provide a list of choices, only one of which has to be satisfied. The example supposes that the tonnages for both *Wheat* and *Soybeans* have been overestimated so if a record refers to one or other of these commodities it must be adjusted. The *logical_test* argument uses the **OR** function to specify that the value in the

Commodity column (cell E2) must be either *Wheat* or *Soybeans*. Note that the Tonnes values in rows 2,4 and 5 of the screenshot have been changed (Fig.3).

G2							fx
							=IF(OR(E2="Wheat",E2="Soybeans"),F2*95%,F2)
	A	B	C	D	E	F	G
1	Year	Month	CountryFrom	CountryTo	Commodity	Tonnes	Adjusted Value
2	2000	1	USA	United Kingdom	Wheat	16182	15372.9
3	2005	2	USA	United Kingdom	Rice	11398	11398
4	2002	6	Mexico	Portugal	Wheat	17497	16622.15
5	2005	3	Canada	Netherlands	Soybeans	15627	14845.65
6	2004	9	Brazil	Italy	Maize	10748	10748

Fig.3 The logical test contains alternative conditions.

Using a Nested IF Function

It isn't always practical or possible to add conditions to an **IF** function by using **AND** and **OR** functions. Another way to deal with more complicated decision-making is to place another **IF** function inside the first one so that depending on the result of the first test, another question has to be answered. This second **IF** is called a *nested IF* function.

This example supposes that tonnages for Wheat were overestimated but by a larger amount prior to Year 2002. The first logical test asks if the value in the *Commodity* column (cell E2) is *Wheat*. If the answer is *True* then there is another **IF** function whose logical test asks if the value in the *Year* column (cell A2) is less than 2002. If it is the *Tonnes* value (cell F2) is reduced by 10%, otherwise it is reduced by 5%. However if the value in the *Commodity* column is not *Wheat* then the *Tonnes* value is left unchanged.

Note that in the screenshot (Fig.4) the tonnes values in rows 2 and 4 have both been changed but by different amounts (an additional *Percentage* column has been added to show the extent of the change).

G2								fx
								=IF(E2="Wheat",IF(A2<2002,F2*90%,F2*95%),F2)
	A	B	C	D	E	F	G	H
1	Year	Month	CountryFrom	CountryTo	Commodity	Tonnes	Adjusted Value	Percentage
2	2000	1	USA	United Kingdom	Wheat	16182	14563.8	90%
3	2005	2	USA	United Kingdom	Rice	11398	11398	100%
4	2002	6	Mexico	Portugal	Wheat	17497	16622.15	95%
5	2005	3	Canada	Netherlands	Soybeans	15627	15627	100%
6	2004	9	Brazil	Italy	Maize	10748	10748	100%

Fig.4 Adding a nested IF function

Excel 2007 permits up to 64 nested **IF** functions. In earlier versions of Excel the limit was 7.

Calculating Age Using Nested If Functions

The examples above are fairly uncomplicated but Excel is capable of solving extremely complex logical calculations and nested IF functions are often used, in conjunction with other logical functions, to do this. One challenge is to accurately calculate a persons age in years, using the current date and the person's date of birth.

How Excel Works with Dates

Excel considers dates as numbers. Each date is assigned a unique *serial number*. For example, the 27th September 2009 has the date serial 40083. Fortunately, you don't need to know this but the fact that all dates have numerical values can be very useful. Windows uses the 1900 date system in which 1st January 1900 is date serial 1, 2nd January 1900 is date serial 2 and so on.

When you type a date into a cell, Excel shows you a date but is thinking of a number. To find out the serial number of a date, select the cell containing the date then open the **Format Cells** dialog box by clicking the arrow in the corner of the **Number** section of the Home tab of the ribbon (in Excel 2003 go to **Format > Cells**). On to the **Number** tab of the dialog box and click **General** in the **Category** list. The date's serial number will appear in the **Sample** box on the right (Fig.5).




Fig.5 Excel stores dates as serial numbers.

You can make use of these numbers in all sorts of ways. You can add a number to a date to give a date that number of days later (or subtract a number to get a date before), you can take one date from another to find out how many days in between. There are lots of ready-made date functions too.

Working Out a Person's Age

A person's age is the amount of time since they were born (I know you know that but the computer doesn't, and we have to start thinking like the computer). So, all we have to do is put today's date in one cell and the person's date of birth in another cell, then take their date of birth away from today and you get their age - right? Well, sort of... you get a number. Because you took a date serial from another date serial you get the number of days in between (Fig.6).




Fig.6 The result of subtracting dates is a number.

We need to convert this number of days into a number of years. Most years have 365 days but every fourth year has 366 days. So the average number of years is 365.25. You can modify the formula by dividing the result of the subtraction by this number (Fig.7). Note the brackets around the first part of the formula. Brackets mean "Work out this bit first...". I've used them here to stop Excel trying to divide A2 by 365.25 before taking it away from A1. Excel formulas work do any multiplying and dividing *before* it does adding and subtracting, but anything in brackets gets done first.




Fig.7 Dividing by the number of days in a year gives the age in years.

Now we can see a number of years, but it's still not quite right. We are getting an accurate result but we don't really want to see the fraction. As a last refinement we'll wrap the whole thing inside an **INT()** function to give us a whole number (an *integer*). This is better than changing the number of decimal places displayed, which would risk some numbers being rounded up and giving an incorrect result. In the finished result (Fig. 8) the formula in cell **A3** is: **=INT((A1-A2)/365.25)**




Fig. 8 Using the Integer function to round the value down to a whole number.

Note: Sometimes when working with dates in Excel you get a result in the form of a date when you were expecting a number. Excel is trying to help but has misunderstood what you needed. In calculations involving already formatted cells, the result cell is usually automatically formatted the same way as the first cell in the formula. Because the first cell was formatted as a date Excel showed you the result as a date, although you wanted to see a number. Just reformat the cell manually by setting its number format to **General**.

How Accurate Do You Need to Be?

This formula yields pretty accurate results but it isn't infallible. Dividing by the average number of days in a year works for most people most of the time, but sometimes it gets it wrong. Supposing the person in question is a child, who hasn't yet lived through a 366 day year, you should be dividing by 365 and not 365.25. So how can we get an exact, guaranteed correct figure? Read on...

Using Nested IF Statements to Tell It Like It (Really) Is!

Supposing it's August and you need to know old someone is...

- The person was born in 1975. How old are they? You can't say.
- The person was born in August 1975. How old are they? You still can't say.
- A person was born on 23rd August 1975. How old are they? At last! You have enough information. You can say for certain.

In order to calculate someone's age precisely you need to know the year in which they were born, and whether or not they have had their birthday. If they have had their birthday you subtract their birth year from the current year. If they have not yet had their birthday you subtract their birth year from the current year, and then subtract 1. Easy! We do it all the time without thinking about it. But explaining the rules to Excel is a bit more complicated. Here goes...

```
=IF(MONTH(TODAY())>MONTH(A1),YEAR(TODAY())-YEAR(A1),
IF(AND(MONTH(TODAY())=MONTH(A1),DAY(TODAY())>=DAY(A1)),
YEAR(TODAY())-YEAR(A1),(YEAR(TODAY())-YEAR(A1))-1))
```

I've written this calculation on three lines for clarity but you should write it as a single expression without spaces. It assumes that cell A1 contains the person's date of birth. Here's what it says...

- **IF(MONTH(TODAY())>MONTH(A1))**
If this month is later than the month of the person's birthday...
- **YEAR(TODAY())-YEAR(A1)**
...subtract the year in which they were born from this year because they must have had their birthday.

But what if we haven't passed the month in which they were born. We might be in that month, or we might not have reached it yet. Let's find out...

- **IF(AND(MONTH(TODAY())=MONTH(A1),DAY(TODAY())>=DAY(A1)))**
If we are currently in the month of the person's birthday **and** it is either their birthday today or we have passed it...
- **YEAR(TODAY())-YEAR(A1)**
...subtract the year in which they were born from this year because they must have had their birthday.

But what if this isn't the month in which they were born. We know we haven't passed their birthday so...

- **(YEAR(TODAY())-YEAR(A1))-1**
...subtract the year in which they were born from this year then subtract 1, because they haven't had their birthday yet.

Here is the result in Excel:

	A	B	C	D	E	F	G	H	I	J
1	14/09/2009	Today								
2	27/09/1950	Date of Birth								
3	58	Age								
4										
5										

Fig.9 An accurate age calculated using nested If functions.